Using Finite-Dimensional Complementarity Problems to Approximate Infinite-Horizon Optimization Models

Thomas F. Rutherford

October 13, 2005

The Ramsey model may be defined as:

$$
\begin{aligned}
&\max \quad \tfrac{1}{1-\eta} \textstyle\sum_{t=0}^{\infty} \beta_t C_t^{1-\eta} \\
&s.t. \\
&\quad C_t = a K_t^b \bar{\ell}_t^{1-b} - I_t \\
&K_{t+1} = (1-\delta) K_t + I_t \\
&K_0 = \bar{K}_0
\end{aligned}
\tag{1}
$$

The problem with representing this model using conventional optimization software is the infinite horizon. Numerical solution algorithms for optimization are normally limited to finite-dimensional models.

1 A Naive Approach

One approach one might consider would be to ignore the infinite horizon and simply formulate the model as a finite-dimensional nonlinear program. This type of model is easily implemented, as illustrated by the following GAMS code:

```
variables
        C(t)   Consumption
        K(t)   Capital stock
        Y(t)   Production
        I(t)   Investment
        W      Intertemporal utility;

equations
        production  Cobb Douglas production function
        allocation  Output market
```

```
         accmulation Capital stock evoluation
         utility     Intertemporal walfare;

utility..          W =e= sum(t, beta(t) * C(t)**(1-eta)) /(1-eta);

production(t)..    Y(t) =e= a * K(t)**b * l(t)**(1-b);

allocation(t)..    Y(t) =g= C(t) + I(t);

accmulation(t+1).. K(t+1) =e= (1-delta)*K(t) + I(t);
```

Unfortunately, the finite formulation shown here has limited usefulness for practical applications due to "terminal effects". The optimal solution to this model involves zero investment for many of the final periods, as the value of the capital stock falls to zero in the final year.

We can illustrate this phenomenon by setting up a calibrated numerical example in which we assume the base year ($t = 0$) is consistent with the long-run steady-state. This implies the following assignment of input parameters such that the model is consistent with a stationary growth path:

```
*       Benchmark data assumnptions:

scalars
        g       Labour growth rate in efficiency units  /0.023/
        delta   Capital depreciation rate               /0.04/
        i0      Base year investment  /0.30 /
        c0      Base year consumption /0.27 /
        b       Capital value share /0.65 /

*       Calibrated parameters:

scalars
        rho     Calibrated marginal product of capital,
        k0      Initial capital
        l0      Initial labour
        a       Cobb Douglas scale parameter;

*       Calibrate base year investment:

k0  = i0/(g+delta);

*       Calibrate the base year marginal product of capital:

rho = b*(c0+i0)/k0-delta;

*       Interrupt the solution process if the model is unbounded:

abort$(g > rho)  "Growth rate exceeds discount rate.",g,rho;

*       Labor supply in the base year:
```

```
l0  = (1-b)*(c0+i0);

*       Labor supply over the model horizon:

l(t) = l0 * power(1+g, ord(t)-1);

*       Utility discount parameter:

beta(t)= power((1+g)**eta/(1+rho), ord(t)-1);

*       Calibration of production scale parameter:

a = (c0+i0)/(k0**b * l0**(1-b));
```

The model is then initialized with the steady-state capital stock, i.e.

```
K.fx(tfirst) = k0;
```

This specification begins right at the steady-state equilibrium for which the optimal policy involves investment in each period at a level which covers growth plus depreciation, i.e.

$$I_t^* = (g + \delta)K_t^* \quad \forall t \geq 0$$

While this is the "true solution", the numerical solution fails to take into account the post-terminal return to capital, and it therefore optimizes the finite-horizon utility by permitting investment fall to zero during the final decade of the planning horizon. Figure 1 compares model outputs for horizons varying from 60 to 200 years. In each of these simulations, the optimal value of investment in the terminal periods is zero, and the model fails to replicate the known infinite-horizon policy regime, particularly for years more than 10 years into the future.

Figure 1 goes about here.

2 Simple Terminal Constraints may be Counterproductive

Seeing that the problem with terminal approximation is the representation of steady-state investment levels in the final period, one might consider simply imposing a constraint requiring that terminal investment be of sufficient level to cover both growth and depreciation of capital stocks in the final period, i.e.

$$I_T = (g + \delta)K_T$$

In the GAMS model this is written:

```
terminvest(tlast)..     I(tlast) =g= (g+delta) * K(tlast);
```

This simple constraint is unable to correct for the terminal approximation error. In fact, in many cases it tends to exaccerbate the problem, as illustrated in Figure 2. Here we see that the introduction of a terminal investment target tends to cause investment to fall to zero for a period of several years prior to the terminal period. This leads to a lower level of aggregate capital in the terminal period, reducing the investment costs which otherwise would be introduced through the terminal investment constraint.

Figure 2 goes about here.

3 An NLP-Based Approximation Method

The issue of how to approximate an infinite horizon programming problem with a finite-dimensional model was a research topic in the 1960's. Among a number of papers (see, e.g. Eckhaus and Parikh (1968) and Manne (1970)), I find the paper by Barr and Manne (1967) to be particularly clear and intuitive. The approach applies a constraint on terminal investment as a fraction of the terminal capital stock. In addition, Barr and Manne propose an adjustment of the weight placed on terminal period consumption. This increased weight accounts for consumption in the post-terminal period, the effect of which is to diminish the incentive to drive investment to zero in the years leading up to the terminal period.

The adjustment factor is calculated as the present-value index of a consumption index which grows at rate g with interest rate ρ from the final period of the model to the infinte horizon, i.e.

$$\phi = \sum_{t=T}^{\infty} \left(\frac{1+g}{1+\rho}\right)^{t-T} = \frac{1+r}{r-g}$$

In the GAMS program, this adjustment of β_T is performed with the following assigment:

```
beta(tlast) = beta(tlast) * (1+rho)/(rho-g)
```

After having made these changes (terminal investment constraint plus increased weight on terminal period consumption) the nonlinear programming model faithfully reproduces the steady-state growth path when initiated at the terminal point.

4 The Complementarity Problem

A complementarity problem corresponding to a nonlinear program is a system of Kuhn-Tucker conditions. These types of model can be represented in most modern modelling languages, and they offer a potential improvement over the Barr-Manne approximation method for approximateing long-term adjustment paths with short-term model horizons.

Let us first look at what a complementarity problem looks like in this instance. The MCP model shown below nearly corresponds equation-by-equation to the first-order conditions for the Manne-Barr model, however in this model we incorporate one additional variable (the salvage value of capital immediate following the final year of the model) and a corresponding termination condition which either relates terminal investment to the terminal capital stock or it targets the *growth rate* of investment in the terminal two years of the model.

```
variables
         C(t)   Consumption
         K(t)   Capital stock
         Y(t)   Production
         I(t)   Investment

         PY(t)  Shadow value of output,
         P(t)   Shadow value of market supply,
```

```
        PK(t)  Shadow value of capital,
        PKT(t) Salvage value of capital;

*       Primal constraints (from the NLP):

production(t)..         a * K(t)**b * l(t)**(1-b) =E= Y(t);

allocation(t)..         Y(t) =g= C(t) + I(t);

accmulation(t)..        (1-delta)*K(t-1) + I(t-1) + (k0*%k0%)$tfirst(t) =e= K(t);

*       Dual constraints (first-order-conditions from the NLP):

foc_y(t)..      PY(t) =E= P(t);

foc_c(t)..      P(t) =E= beta(t) * C(t)**(-eta) / (c0*qref(t))**(1-eta);

foc_i(t)..      P(t) =G= PK(t+1) + PKT(t)$tlast(t);

foc_k(t)..      PK(t) =E= b * PY(t) * a * K(t)**(b-1) * l(t)**(1-b)

                        + (1-delta) * (PK(t+1)+PKT(t)$tlast(t));

*       Terminal approximation constraint (non-integrable equation):

kt(tlast)..     sum(t$tlast(t), I(t) - (1+g) * I(t-1))$svtarget +

                (I(tlast) - (g+delta) * K(tlast))$(not svtarget =E= 0;
```

We have argued (Lau, Pahlke and Rutherford 2002) that the “state-variable targetting” approach provides more precise terminal approximiations than Manne-Barr. This is true, but the improvement in precision depends on the application. In the present model, the state-variable method leads to a modest improvement in the terminal approximation, as is illustrated in Figure 3.

In this experiment, we contemplate an unanticipate drop in the capital stock and subsequently we trace out the adjustment path. (See model input `k0` which is interpreted as a scale factor on base year capital stock relative to the steady-state value, `k0=1`.)

This adjustment path involves a short-term decrease in investment which gradually returns to the long-run steady state over a period of more than 100 years. In our simulations, we evaluate the terminal approximation over time intervals of 30, 60 and 200 years, using both the Manne-Barr NLP formulation and the state-variable targetting MCP formulation. We see that for short time horizons, the MCP formulation with targeting of terminal investment generally performs better than the NLP approach. The reason we get a more precise approximation is that it is easier to characterize the terminal *growth rate* of investment rather than the *level* of investment in the final year.

Figure 3 goes about here.

References

[1] Barr, J. R. and A. S. Manne, “Numerical Experiments With Finite Horizon Planning Models,” Indian Economic Review, April 1967, 1-29.

[2] Manne, A. S., “Sufficient Conditions for Optimality in an Infinite Horizon Development Plan,” Econometrica, 38(1), January 1970, 18-38.

[3] Eckhaus, R. S. and K. S. Parikh, Planning for Growth (Cambridge, MA: MIT Press, 1968).

[4] Lau, M. I., A. Pahlke and T. F. Rutherford, “Approximating infinite-horizon models in a complementarity format: a primer in dynamic general equilibrium analysis,” Journal of Economic Dynamics and Control 26(4):577–609, 2002.



Figure 1 Investment Approximation Errors for Alternative Horizon Dates
(Ramsey Model with No Terminal Constraints)



Figure 2 Investment Approximation Errors for Alternative Horizon Dates
(Ramsey Model with Terminal Constraint: $I_T = (g + \delta) K_T$



Figure 3 Investment Response to Unanticipated 50% Loss in Capital Stock (NLP=Manne-Barr Model - MCP=State Variable Targetting)